

Mengetahui MAC Address Pada Komputer Anda Dengan Visual Basic 6.0

Sony Arianto Kurniawan

sony-ak@sony-ak.com
Sony AK Knowledge Center
<http://www.sony-ak.com/>

Lisensi Dokumen:

Copyright © 2003 IlmuKomputer.Com

Seluruh dokumen di IlmuKomputer.Com dapat digunakan, dimodifikasi dan disebarkan secara bebas untuk tujuan bukan komersial (nonprofit), dengan syarat tidak menghapus atau merubah atribut penulis dan pernyataan copyright yang disertakan dalam setiap dokumen. Tidak diperbolehkan melakukan penulisan ulang, kecuali mendapatkan ijin terlebih dahulu dari IlmuKomputer.Com.

Suatu hari teman kuliah penulis mengirim e-mail yang berisi pertanyaan mengenai masalah Visual Basic 6.0. Dia ingin mengetahui MAC address pada ethernet card komputernya dengan menggunakan Visual Basic. Gimana caranya? OK, simak saja tulisan berikut.

Sebelumnya mari kita cari tahu dulu sebenarnya apa sih yang dimaksud dengan MAC address itu. MAC (Medium Access Control) address adalah suatu angka yang sifatnya unik yang terdiri dari 48 bit. Angka ini diperuntukkan bagi NIC (Network Interface Card) dan dikeluarkan oleh vendor dari NIC tersebut. MAC address ini (yang bisa dikatakan sebagai physical address) digunakan untuk keperluan mapping pada komunikasi jaringan yang berbasis TCP/IP.

Untuk informasi lebih lanjut mengenai MAC address ini Anda bisa cari di internet. Kali ini langsung saja kita masuk pada misi utama. Silakan Anda buka Visual Basic 6.0 dan pada form kosong yang sudah tersedia tambahkan sebuah Text Box (sebagai tempat untuk menampilkan MAC address nantinya) dan sebuah kontrol Command Button (sebagai pemicu untuk menjalankan fungsi mendapatkan nilai MAC address).

Nama kontrol Command Button pada Form1 secara default adalah Command1. Sekarang tambahkan kode dibawah ini pada bagian deklarasi umum di Form1.

```
Option Explicit
```

```
Private Const NCBASTAT = &H33  
Private Const NCBNAMSZ = 16  
Private Const HEAP_ZERO_MEMORY = &H8  
Private Const HEAP_GENERATE_EXCEPTIONS = &H4  
Private Const NCBRESET = &H32
```

```
Private Type NCB
```

```
ncb_command As Byte 'Integer
ncb_retcode As Byte 'Integer
ncb_lsn As Byte 'Integer
ncb_num As Byte ' Integer
ncb_buffer As Long 'String
ncb_length As Integer
ncb_callname As String * NCBNAMSZ
ncb_name As String * NCBNAMSZ
ncb_rto As Byte 'Integer
ncb_sto As Byte ' Integer
ncb_post As Long
ncb_lana_num As Byte 'Integer
ncb_cmd_cplt As Byte 'Integer
ncb_reserve(9) As Byte ' Reserved, must be 0
ncb_event As Long
End Type
Private Type ADAPTER_STATUS
adapter_address(5) As Byte 'As String * 6
rev_major As Byte 'Integer
reserved0 As Byte 'Integer
adapter_type As Byte 'Integer
rev_minor As Byte 'Integer
duration As Integer
fmr_rcv As Integer
fmr_xmit As Integer
iframe_rcv_err As Integer
xmit_aborts As Integer
xmit_success As Long
rcv_success As Long
iframe_xmit_err As Integer
rcv_buff_unavail As Integer
tl_timeouts As Integer
ti_timeouts As Integer
Reserved1 As Long
free_ncbs As Integer
max_cfg_ncbs As Integer
max_ncbs As Integer
xmit_buf_unavail As Integer
max_dgram_size As Integer
pending_sess As Integer
max_cfg_sess As Integer
max_sess As Integer
max_sess_pkt_size As Integer
name_count As Integer
End Type
Private Type NAME_BUFFER
name As String * NCBNAMSZ
name_num As Integer
name_flags As Integer
End Type
Private Type ASTAT
adapt As ADAPTER_STATUS
NameBuff(30) As NAME_BUFFER
End Type

Private Declare Function Netbios Lib "netapi32.dll" _
    (pncb As NCB) As Byte
Private Declare Sub CopyMemory Lib "kernel32" Alias "RtlMoveMemory" ( _
    hpvDest As Any, ByVal hpvSource As Long, ByVal cbCopy As Long)
```

```
Private Declare Function GetProcessHeap Lib "kernel32" () As Long
Private Declare Function HeapAlloc Lib "kernel32" _
    (ByVal hHeap As Long, ByVal dwFlags As Long, _
    ByVal dwBytes As Long) As Long
Private Declare Function HeapFree Lib "kernel32" (ByVal hHeap As Long, _
    ByVal dwFlags As Long, lpMem As Any) As Long
```


Kemudian tambahkan code berikut ini pada event Click pada kontrol CommandButton.

```
Private Sub Command1_Click()
    Dim myNcb As NCB
    Dim bRet As Byte
    myNcb.ncb_command = NCBRESET
    bRet = Netbios(myNcb)

    myNcb.ncb_command = NCBASTAT
    myNcb.ncb_lana_num = 0
    myNcb.ncb_callname = "*"

    Dim myASTAT As ASTAT, tempASTAT As ASTAT
    Dim pASTAT As Long
    myNcb.ncb_length = Len(myASTAT)
    Debug.Print Err.LastDllError
    pASTAT = HeapAlloc(GetProcessHeap(), HEAP_GENERATE_EXCEPTIONS _
        Or HEAP_ZERO_MEMORY, myNcb.ncb_length)
    If pASTAT = 0 Then
        Debug.Print "memory allocation failed!"
        Exit Sub
    End If
    myNcb.ncb_buffer = pASTAT
    bRet = Netbios(myNcb)
    Debug.Print Err.LastDllError
    CopyMemory myASTAT, myNcb.ncb_buffer, Len(myASTAT)
    Text1.Text = Hex(myASTAT.adapt.adapter_address(0)) & " " & _
        Hex(myASTAT.adapt.adapter_address(1)) _
        & " " & Hex(myASTAT.adapt.adapter_address(2)) & " " _
        & Hex(myASTAT.adapt.adapter_address(3)) _
        & " " & Hex(myASTAT.adapt.adapter_address(4)) & " " _
        & Hex(myASTAT.adapt.adapter_address(5))
    HeapFree GetProcessHeap(), 0, pASTAT
End Sub
```

Kemudian coba jalankan program tersebut dengan menekan F5. Screen shot dari contoh program ini bisa dilihat seperti pada gambar berikut.



OK sekian dulu tulisan berikut. Thanks buat rekan Akhmad Arief Ulinnuha atas request mengenai artikel ini. Jika Anda mempunyai komentar, saran atau pertanyaan silakan dilayangkan ke sony-ak@sony-ak.com. Disertakan juga source code contoh dari tulisan ini.

Source diperoleh dari Microsoft Q175472

<http://support.microsoft.com/default.aspx?scid=kb;en-us;175472>

Sekian dan terimakasih.

Terimakasih.

Sony Arianto Kurniawan,

Founder and Creative Director of The Sony AK Knowledge Center

sony-ak@sony-ak.com

www.sony-ak.com